1 Local Binding Expressions

Studying the Intex language is able to give us insight into fundamental programming processes like interpretation and program analysis. But Intex is missing many important features of real programming languages. Our goal is to explore various programming language dimensions by incrementally extending Intex with simple versions of various features, and then to explore the design space associated with such features.

The first feature we shall explore is the ability to name values. We shall do this in a new language, Bindex, an extension to Intex in which:

1. Program arguments are referenced by name rather than by position. For example, in Bindex, the averaging program can be written as:

   (bind (x y) (/ (+ x y) 2))

   The program inputs ((x y) in this case) are called the **formal parameters** of the program and each use of such a parameter in the program is called a **variable reference**.

2. There is a new kind of expression that has the form:\(^1\)

   (bind \(I\) \(name\) \(E\) \(defn\) \(E\) \(body\))

   Intuitively, a bind expression is evaluated as follows:

   (a) The **definition expression** \(E_{defn}\) is evaluated to an integer value \(V_{defn}\).

   (b) The **body expression** \(E_{body}\) is evaluated in such a way that each variable reference \(I_{name}\) denotes \(V_{defn}\).

   (c) The integer value of \(E_{body}\) is returned as the value of the bind expression.

Here is a simple example of a Bindex program using bind to calculate the value \(\frac{a^2 + b^2}{a^2 - b^2}\).

   (bind (a b)
          (bind a_sq (* a a)
                 (bind b_sq (* b b)
                    (bind numer (+ a_sq b_sq)
                       (bind denom (- a_sq b_sq)
                          (/ numer denom)))))))

Constructs like `bind` in Bindex, `let` in OCaml and in Scheme, and `type varname = exp;` in Java and C are known as **local binding expressions** or **local variable declarations**. A local binding expression introduces a name for the result of a definition expression that can be used in some part of the program (in the case of `bind`, the body of the `bind` expression). Local binding expressions are used for three main purposes in a program:

\(^1\)We shall use (potentially subscripted) occurrences of the metavariable \(I\) to stand for names (a.k.a. identifiers) in a language and (potentially subscripted) occurrences of the metavariable \(E\) to stand for expressions.
1. Naming the result of evaluating a definition expression can avoid the cost of recalculating the value of that expression. For instance, in the above example, naming the result of \((a \ a a)\) means that the multiplication needs to be performed only once; it would need to be performed twice if the result were not named. In the case of a simple expression like \((a \ a a)\), it is not clear that any time is saved by avoiding recalculation. However, for expensive calculations, the time saved by avoiding recalculation can be considerable. In the context of recursion, naming the result of a recursive call can often change the asymptotic complexity of an algorithm.

2. In programs where values can have a time-varying state (think Java objects), it is essential to name values so that the same value can be referenced more than once. Since it is stateless, this is not important in BINDEX, but it will be important for later extensions to BINDEX that support stateful values.

3. Even when a result is not used more than once, naming the result of an intermediate expression can make a program easier to read. For instance, in the above example, \texttt{numer} and \texttt{denom} do not make the calculation any more efficient, but some programmers might find the code more readable than without the names. Such naming can be especially handy for breaking down deeply nested expressions into more manageable subexpressions.

## 2 BINDEX Syntax

The abstract syntax for BINDEX (Fig. 1) is the same as that for INTEX except for three changes:

1. The program form \texttt{Pgm} replaces the number of program arguments by a list of formal parameter names.

2. The argument reference form \texttt{Arg of int} is replaced by the variable reference form \texttt{Var of var}, where \texttt{var} is a synonym for \texttt{string}.

3. There is a new form, \texttt{Bind of var * exp * exp}, that represents the \texttt{bind} expression. The \texttt{var} component is the \texttt{bound variable}, while the two subexpressions, in order, are the \texttt{definition} (whose value is bound to the variable) and the \texttt{body} in which the bound variable may be used.

```ocaml
type var = string

type pgm = Pgm of string list * exp (* param names, body *)

and exp =
  Lit of int (* integer literal with value *)
| Var of var (* variable reference *)
| BinApp of binop * exp * exp (* binary operator application with rator, rands *)
| Bind of var * exp * exp (* bind name to value of defn in body *)

and binop =
  | Add | Sub | Mul | Div | Rem (* binary arithmetic ops *)
```

Figure 1: Abstract syntax for BINDEX expressed in OCAML.

A \texttt{bind} expression may be used anywhere an expression is expected, including the rand positions of a binary application and the definition of another \texttt{bind} expression. For example, here is a sample program illustrating the flexibility of \texttt{bind} expressions (Fig. 2 shows its AST):
\[
\text{let rec fold litfun varfun appfun bindfun exp =}
\text{match exp with}
| \text{Lit i} \rightarrow \text{litfun i} \\
| \text{Var s} \rightarrow \text{varfun s} \\
| \text{BinApp(rator, rand1, rand2)} \rightarrow
\text{appfun rator}
\text{(fold litfun varfun appfun bindfun rand1)}
\text{(fold litfun varfun appfun bindfun rand2)}
| \text{Bind(name, defn, body)} \rightarrow
\text{bindfun name}
\text{(fold litfun varfun appfun bindfun defn)}
\text{(fold litfun varfun appfun bindfun body)}
\]

However, when fold functions get this complex, we tend to use explicit pattern matching with recursion to process such ASTs.

The Bindex module contains the definition of the BINDEX abstract syntax along with various utilities for manipulating the syntax. These include the following parsing and unparsing functions:

val sexpToPgm : Sexp.sexp -> Bindex.pgm
val sexpToExp : Sexp.sexp -> Bindex.exp
val stringToPgm : string -> Bindex.pgm
val stringToExp : string -> Bindex.exp
val pgmToSexp : Bindex.pgm -> Sexp.sexp
val expToSexp : Bindex.exp -> Sexp.sexp
val pgmToString : Bindex.pgm -> string
val expToString : Bindex.exp -> string

To treat BINDEX as an extension to INTEX, the sexpToPgm function parses a representation of an INTEX program with \(n\) parameters to a BINDEX programs by using \(\$1 \ldots \$n\) as the parameter names and treating each argument reference as if it were the variable reference \(\$i\). For instance, an INTEX averaging program is parsed as if it were written

\[
(bindex (\$1 \$2) (/ (+ \$1 \$2) 2))
\]

This translation is sound as long as no local variable names have the form \(\$i\).
Figure 2: Abstract syntax trees for the sample BINDEX program.
3 General Naming Issues

Before we study how the bind construct is evaluated, we first consider some general concepts and issues related to naming in programming languages and mathematics.

3.1 Declarations and Scope

Programming languages and mathematical languages almost always have constructs that introduce names for the kinds of entities that are manipulated by the language. Such constructs are known as declarations or binding constructs. Figure 3 shows some examples of declarations from programming and mathematics with which you are probably familiar.

Every declaration construct has a binding occurrence that introduces the declared name, and reference occurrences that refer to declared name. For example, in the Ocaml abstraction `fun x -> x * x`, the first `x` is the binding occurrence, and the second and third `x`s are reference occurrences. Typically, the binding occurrence and reference occurrences have the same syntax; they are distinguished by their positions within the declaration construct. So in `fun`, for instance, the name following the `fun` keyword is the the binding occurrence, and the uses of this name in the body are reference occurrences.

Once declared, a name can usually only be used within a restricted part of the program. The region of a program in which it is possible to reference a declared name is called the scope of the declared name. In so-called statically scoped languages, the scope of declared names can be shown via nested boxes called lexical contours. For example, Fig. 4 shows the lexical contours for the sample Bindex program. The scope of the formal parameter names `x` and `y` is the entire body of the program (contour `C_0`). The scope of a local name introduced by a `bind` encompasses the body of the `bind` expression but does not include the definition expression. This can be seen by the contours for `a` (`C_1`), `b` (`C_1`), `c` (`C_3`), and `d` (`C_4`).

When contours are nested, a name declared in an outer contour may be used within an inner contour unless the inner contour declares the same name as the outer contour. For example, the names which may be used in context `C_2` are `x`, `y`, `a`, and `b`. However, suppose we rename `b` to `x` so that the first `bind` becomes

```
(bind a (/ y x)
  (bind x (- a y)
   (* a x)))
```

In this case, the `x` that is referenced in `C_2` refers to the local name `x` declared in `C_2`, not the program parameter `x` declared in `C_0`. So the program parameter `x` declared in `C_0` may be references everywhere in the program except in `C_2`. The inner declaration of `x` is said to shadow the outer one, and the contours of the inner declaration is said to be a hole in the scope of the outer declaration.

3.2 Free Variables

In a given program phrase, a reference occurrence of a name for which there is no binding occurrence is called a free variable or unbound variable; otherwise it is said to be a bound variable. For instance:

- in the Bindex expression `(+ a b)`, `a` and `b` are free variables.
- in the Bindex expression `(bind b (* 2 3) (+ a b))`, `a` is a free variable, but `b` is bound.
- in the Bindex expression `(bind a (- 8 1) (bind b (* 2 3) (+ a b)))`, both `a` and `b` are bound.
<table>
<thead>
<tr>
<th>Language</th>
<th>Construct</th>
<th>Example</th>
</tr>
</thead>
<tbody>
<tr>
<td>OCAML</td>
<td>fun I \rightarrow E_{body}</td>
<td>fun x \rightarrow x * x</td>
</tr>
<tr>
<td>OCAML</td>
<td>let I_1 = E_1 and ... and I_n = E_n in E_{body}</td>
<td>let x = 2 + 3 and y = 6 * 7 in let z = y / x in x * (y + z)</td>
</tr>
<tr>
<td>SCHEME</td>
<td>(lambda (I_1 ... I_n) E_{body})</td>
<td>(lambda (a b) (lambda (c) (* b (+ a c))))</td>
</tr>
<tr>
<td>SCHEME</td>
<td>(let ((I_1 E_1) ... (I_n E_n)) E_{body})</td>
<td>(let ((x (+ 2 3)) (y (* 6 7)) (let ((z (quotient y x))) (* x (+ y z))))</td>
</tr>
<tr>
<td>SCHEME</td>
<td>(define I_1 E_{body})</td>
<td>(define a (+ 2 3)) (define square (lambda (x) (* x x))) (define a-squared (square a))</td>
</tr>
<tr>
<td>JAVA</td>
<td>T_{returnType} I_{varname} = E_{defn};</td>
<td>int x = 2 + 3; int y = 6 * 7; int z = y / x;</td>
</tr>
<tr>
<td>BINDEX</td>
<td>(bind I_{name} E_{defn} E_{body})</td>
<td>(bind x (+ 2 3) (bind y (* 6 7) (bind z (/ y x) (* x (+ y z)))))</td>
</tr>
<tr>
<td>Math</td>
<td>\sum_{I_{var} \rightarrow E_{lo}} E_{body}, \prod_{I_{var} \rightarrow E_{lo}} E_{body} \sum_{i=1}^{10} \left( \prod_{j=1}^{i} (i + j) \right)</td>
<td></td>
</tr>
<tr>
<td>Logic</td>
<td>\forall I_{var}.E_{body}, \exists I_{var}.E_{body} \forall x.\exists y.(x + 1) = y</td>
<td></td>
</tr>
<tr>
<td>Calculus</td>
<td>\int_{E_{lo}}^{E_{hi}} E_{body} , dI_{var} \int_{0}^{1} x \cdot (\int_{0}^{x} y , dy) , dx</td>
<td></td>
</tr>
</tbody>
</table>

Figure 3: Examples of declarations in programming and mathematics.
Note that some occurrences of a name in an expression can be free while other occurrences of the same name may be bound. Consider the following expression, in which different reference occurrences of \texttt{a} and \texttt{b} are distinguished by subscript:

\begin{verbatim}
(bind a (- a _{1} b _{1})
  (bind b (* a _{2} b _{2})
    (+ a _{3} b _{3}))
\end{verbatim}

The variable references \texttt{a}_{1}, \texttt{b}_{1}, and \texttt{b}_{2} are free while \texttt{a}_{2}, \texttt{a}_{3}, and \texttt{b}_{3} are bound.

The **free variables** of a program phrase is the set of all variable names that occur free in that phrase. The following table shows how free variables are calculated for \texttt{Bindex} expressions and programs:

<table>
<thead>
<tr>
<th>Bindex Phrase P</th>
<th>Free Variables: FV(P)</th>
</tr>
</thead>
<tbody>
<tr>
<td>\texttt{L} (integer literal)</td>
<td>{}\</td>
</tr>
<tr>
<td>\texttt{I} (variable reference)</td>
<td>{I}</td>
</tr>
<tr>
<td>\texttt{(O} \texttt{rator } E_{\text{rand}1} E_{\text{rand}2})</td>
<td>FV(E_{\text{rand}1}) \cup FV(E_{\text{rand}2})</td>
</tr>
<tr>
<td>\texttt{(bind I } E_{\text{defn}} E_{\text{body}})</td>
<td>FV(E_{\text{defn}}) \cup (FV(E_{\text{body}}) - {I})</td>
</tr>
<tr>
<td>\texttt{(bindex } (I_{1} \ldots I_{n}) E_{\text{body}})</td>
<td>FV(E_{\text{body}}) - {I_{1} \ldots I_{n}}</td>
</tr>
</tbody>
</table>

Fig. 5 shows how these free variable calculations are expressed as the functions \texttt{freeVarsExp} and \texttt{freeVarsPgm} in the \texttt{Bindex} module. The functor invocation \texttt{Set.Make(String)} uses the standard
OCAML set-making functor to make a module for manipulating sets of strings. The resulting set-of-strings module has the \texttt{SET} signature in Fig. 6. Here are some examples of the free variable functions in action:

```ocaml
# open Bindex;;
# setToList (freeVarsExp (stringToExp "(+ a (* b b)))");;
- : Bindex.S.elt list = ["a"; "b"]
# setToList (freeVarsExp (stringToExp "(bind b (* a c) (+ a (* b b)))");;
- : Bindex.S.elt list = ["a"; "c"]
# setToList (freeVarsExp
  (stringToExp "((bind a (- b c) (bind b (* a c) (+ a (* b b))))");;
- : Bindex.S.elt list = ["b"; "c"]
# setToList (freeVarsPgm
  (stringToPgm
   "((bind (b c) (bind a (- b c) (bind b (* a c) (+ a (* b b))))))");
- : Bindex.S.elt list = []
```

A program phrase with no free variables is said to be \texttt{closed}. All \texttt{BINDEX} programs should be closed since a free variable in a program would have no meaning. The \texttt{varCheck} function in Fig. 5 statically checks a program to ensure that it is closed. If it is, it returns the unit value (). But if there are unbound variables, it raises an \texttt{Unbound} exception with the set of free variables. The \texttt{varCheck} function is the analog in \texttt{BINDEX} of \texttt{INTEX}'s \texttt{argCheck} function.

### 3.3 $\alpha$-Renaming

In a statically scoped language, it is always possible to consistently rename a binding occurrence and its corresponding reference occurrences without changing the meaning of a program. Consistent renaming that maintains program meaning is known as \texttt{$\alpha$-renaming}. For example, in

\begin{verbatim}
(bind b (* a a)
 (bind c (+ b a)
  (bind a (* b c)
   (/ (+ a c) (- a b)))))
\end{verbatim}

we can rename \texttt{b} to \texttt{x}, \texttt{c} to \texttt{y}, and the bound \texttt{a} to \texttt{z} to yield

\begin{verbatim}
(bind x (* a a)
 (bind y (+ x a)
  (bind z (* x y)
   (/ (+ z x) (- z y)))))
\end{verbatim}

Note that the free occurrences of \texttt{a} are not renamed.

There is one restriction on the renaming of bound variables. We cannot rename a bound variable to another variable that is free in its scope: this would cause \texttt{variable capture}. For example, we cannot rename \texttt{b} to \texttt{a} above, since the free reference to \texttt{a} in the expression \texttt{(+ b a)} within the body of \texttt{(bind b ...)} would be captured and become bound to the renamed binding occurrence of \texttt{b}.

When renaming a bound variable, it may be necessary to additionally rename other bound variables in its scope to avoid variable capture. For example, we can rename \texttt{c} to \texttt{a} in our example as long as we also rename the \texttt{bind-bound} \texttt{a} (say to \texttt{a.2}):

\begin{verbatim}
(bind b (* a a)
 (bind a (+ b a)
  (bind a.2 (* b a)
   (/ (+ a.2 a) (- a.2 b)))))
\end{verbatim}

As a more complex of \texttt{$\alpha$-renaming}, we reconsider the sample \texttt{BINDEX} program from earlier:
(* val listToSet : S.elt list -> S.t *)
let listToSet strs = foldr S.add S.empty strs

(* val setToListList : S.t -> S.elt list *)
let setToList set = S.elements set

(* val freeVarsPgm : pgm -> S.t *)
(or Returns the free variables of a program *)
let rec freeVarsPgm (Pgm(fmls, body)) =
  S.diff (freeVarsExp body) (listToSet fmls)

(* val freeVarsExp : exp -> S.t *)
(* Returns the free variables of an expression *)
and freeVarsExp e =
  match e with
  | Lit i -> S.empty
  | Var name -> S.singleton name
  | BinApp(_, rand1, rand2) ->
    S.union (freeVarsExp rand1) (freeVarsExp rand2)
  | Bind(name, defn, body) ->
    S.union (freeVarsExp defn) (S.diff (freeVarsExp body) (S.singleton name))

(* val freeVarsExps : exp list -> S.t *)
(* Returns the free variables of a list of expressions *)
and freeVarsExps es = foldr S.union S.empty (map freeVarsExp es)

(* val varCheck : pgm -> unit *)
and varCheck pgm =
  let unbounds = freeVarsPgm pgm
  in if S.is_empty unbounds then
    () (* OK *)
  else
    raise (Unbound (setToList unbounds))

Figure 5: Functions for determining free variables in BINDEX.

(bindex (x y)
  (+ (bind a (/ y x)
      (bind b (~ a y)
       (* a b)))
  (bind c (bind d (+ x y) (* d y))
     (/ c x))))

It is possible to rename a, b, c, and d to one of x or y without changing the meaning of the program:
module type SET = sig

  type 'a set
  val empty : 'a set (* the empty set *)
  val singleton : 'a -> 'a set (* a set with one element *)
  val insert : 'a -> 'a set -> 'a set (* insert elt into given set *)
  val delete : 'a -> 'a set -> 'a set (* delete elt from given set *)
  val isEmpty: 'a set -> bool (* is the given set empty? *)
  val size : 'a set -> int (* number of distinct elements in given set *)
  val member : 'a -> 'a set -> bool (* is elt a member of given set? *)
  val union: 'a set -> 'a set -> 'a set (* union of two sets *)
  val intersection: 'a set -> 'a set -> 'a set (* intersection of two sets *)
  val difference: 'a set -> 'a set -> 'a set (* difference of two sets *)
  val fromList : 'a list -> 'a set (* create a set from a list *)
  val toList : 'a set -> 'a list (* list all set elts, sorted low to high *)
  val toSexp : ('a -> Sexp.sexp) -> 'a set -> Sexp.sexp (* translates set into s-expression rep. *)
  val fromSexp : (Sexp.sexp -> 'a) -> Sexp.sexp -> 'a set (* translates s-expression rep. into set *)
  val toString : ('a -> string) -> 'a set -> string (* returns string representation of set *)

end

Figure 6: Set module signature.

(bindex (x y)
  (+ (bind x (/ y x)
      (bind y (- x y)
        (* x y)))
    (bind y (bind x (+ x y) (* x y))
      (/ y x)))))

The renamed program may be very difficult for human beings to understand, but because it has
the same wiring diagram as the original program, it is indistinguishable from the original program
for the purposes of program manipulation.

The Bindex module provides two renaming functions:

1. val rename1 : var -> var -> exp -> exp
   rename1 oldName newName e returns a copy of the expression e in which all free occurrences
   of oldName have been renamed to newName.

2. renameAll : var list -> var list -> exp -> exp
   Assume that oldNames and newNames are string lists with the same length. Then the
   invocation renameAll oldNames newNames e returns a copy of the expression e in which all
   free occurrences of names in oldNames have been renamed to the corresponding name (by
   position) in newNames.

We will see in the next section how these functions are implemented. Note that both functions
rename only the free occurrences of a variable in an expression. Program manipulation functions
should not refer to the bound occurrences of variables by name because their behavior should not
changed if the bound variables are α-renamed.

Below are some sample invocations of these functions:
# let testRename1 oldVar newVar expString =
  print_string(expToString(rename1 oldVar newVar (stringToExp expString)));
val testRename1 : Bindex.var -> Bindex.var -> string -> unit = <fun>

# testRename1 "a" "b" "(+ a (bind b (* a a) (+ a b)))";;
(+ b (bind b.10 (* b b) (+ b b.10)))- : unit = ()

Note that the bind-bound occurrences of b have been automatically renamed (to b.10) to avoid variable capture involving the renamed b.

# testRename1 "a" "z" "(+ a (bind b (* a a) (+ a b)))";;
(+ z (bind b.11 (* z z) (+ z b.11)))- : unit = ()

Here the bind-bound b has been renamed (to b.11) even though there is no threat of variable capture. This is an artifact of the way that rename1 is implemented — it automatically renames all bind-bound names to avoid any possibility of variable capture, regardless of whether or not variable capture will actually occur.

# testRename1 "a" "z" "(+ a (bind a (* a a) (+ a a)))";;
(+ z (bind a.12 (* z z) (+ a.12 a.12)))- : unit = ()

This example shows that only the free occurrence of a are renamed to z. The bound occurrences of a are renamed to a.12 as an artifact of the way rename1 is implemented.

# let testRenameAll oldVars newVars expString =
  print_string(expToString(renameAll oldVars newVars (stringToExp expString)));
val testRenameAll : string list -> Bindex.var list -> string -> unit = <fun>

# testRenameAll ["a";"b"] ["b";"a"] "(+ a b)";
(+ b a)- : unit = ()

In this example, note that the renamings are performed simultaneously. This simultaneous renaming cannot be simulated by any ordering of two calls to rename1.

# testRenameAll ["a";"b"] ["b";"a"] "(bind a (+ a b) (bind b (- a b) (* a b)))";;
(bind a.13 (+ b a) (bind b.15 (- a.13 a) (* a.13 b.15)))- : unit = ()

We will use the StringUtils.fresh function to automatically generate “fresh” variable names as needed in our program manipulation functions. This function simply adds a dot followed by a unique number to the given variable name. It is assumed that the original user program does not contain such dotted variable names; they are only introduced by the program manipulation process. For example:

#StringUtils.fresh "foo";;
- : string = "foo.0"
#StringUtils.fresh "foo";;
- : string = "foo.1"
#StringUtils.fresh "bar";;
- : string = "bar.2"
#StringUtils.fresh "foo.0";;
- : string = "foo.3"

The final example shows that when fresh is called on names generated by fresh, only the “root” of the name (before the dot) is used. Thus, fresh "foo.0" yields "foo.3" rather than "foo.0.3". This convention helps to make programs more readable.

We will say that a program is uniquely named if all logically distinct variables in the program have different names. Fig. 7 presents a functions uniquifyPgm and uniquifyExp that transform
BIND EX programs and expressions, respectively, to uniquely named form. This is achieved by using fresh and rename1 to give a unique name to every bind-bound name. The program formal parameters are not renamed; it is assumed that these are all distinct, and renaming the bind-bound names will make them all different from the formal parameters. For example:

```ocaml
let rec uniquifyPgm pgm =  
  match pgm with  
    Pgm(args, body) -> Pgm(args, uniquifyExp body)
and uniquifyExp exp =  
  match exp with  
    Lit i -> exp  
  | Var v -> exp  
  | BinApp(op, r1, r2) -> BinApp(op, uniquifyExp r1, uniquifyExp r2)  
  | Bind(name, defn, body) ->  
    (* rename every bind-bound name to a fresh name *)  
    let name' = StringUtils.fresh name in  
    Bind(name',  
      uniquifyExp defn,  
      uniquifyExp (rename1 name name' body))
```

Figure 7: Functions that transform BINDEX programs and expressions to uniquely named form.

```ocaml
# print_string  
  (pgmToString  
    (uniquifyPgm  
      (stringToPgm  
        "(bind x y)  
          (+ (bind x (/ y x)  
              (bind y (- x y))  
              (* x y)))  
          (bind y (bind x (+ x y) (* x y))  
              (/ y x)))")));;
(bindex (x y)  
  (+ (bind x.5 (/ y x)  
    (bind y.7 (- x.5 y) (* x.5 y.7)))  
    (bind y.3 (bind x.4 (+ x y) (* x.4 y))  
        (/ y.3 x))))- : unit = ()
```

3.4 Substitution

Renaming is a special case of a more general program form of program manipulation called substitution. In substitution, all free occurrences of a variable name are replaced by a given expression. For example, substituting (+ b c) for a in (* a a) yields (* (+ b c) (+ b c)). If we view expressions as trees, then substitution replaces certain leaves (those with the specified name) by new subtrees. For example, substituting the tree

```
BinApp
    rator rand1 rand1
    Add Var Var
    name b name c
```
As in renaming, substitution must avoid variable capture to preserve the “naming wiring structure” of expressions. In particular:

- When substituting for a variable \( I \), we will never perform substitutions on \( I \) within the scope of a \texttt{bind} expression that binds \( I \). For example, substituting \( (+ b c) \) for \( a \) in

\[
\text{bind } a ( \ast a a ) (- a 3)
\]

yields:

\[
\text{bind } a ( \ast (+ b c) (+ b c)) (- a 3)
\]

Here the free occurrences of \( a \) have been replaced by \((+ b c)\), but the bound occurrences have not been changed.

- \texttt{bind}-bound variables may be renamed to avoid variable capture with free variables in the expression being substituted for the variable. For example, substituting \((+ b c)\) for \( a \) in

\[
(+ (\text{bind } b (\ast 1 a) (\ast a b))
\]

\[
(\text{bind } c (\ast 2 a) (\ast a c)))
\]

yields

\[
(+ (\text{bind } b.1 (\ast 1 (+ b c)) (\ast (\ast b c) b.1))
\]

\[
(\text{bind } c.2 (\ast 2 (+ b c)) (\ast (+ b c) c.2)))
\]

Here is is necessary to rename the \texttt{bind}-bound \( b \) and \( c \) to avoid capturing the free variables \( b \) and \( c \) in \((+ b c)\).

Substitution is an important tool for program manipulation. We have already encountered substitution in the context of the substitution model for the evaluation of OCAML programs, where function application was explained by substituting the argument values for the formal parameters in the function body. Later, we will present a similar substitution model for BINDEX evaluation.

\[2\text{In the BINDEX implementation of substitution, the bound } a \text{ will be renamed to a fresh variable in this example.}\]
We will now develop a substitution function for BINDEX that we will be able to use later for manipulating BINDEX programs. The substitution function will allow us to simultaneously substitute expressions for any number of variables. To specify the association between the variable names and the expressions to be substituted for them, we will use environments having the ENV signature in Fig. 8. This is similar to the MENV signature presented in the Modules handout (#17), except that ENV provides a few extra functions (bindAll, remove, removeAll).

Fig. 9 defines the following subst function:

```ocaml
code

module type ENV = sig

  type 'a env

  val empty: 'a env (* returns the empty env *)

  val bind : string -> 'a -> 'a env -> 'a env
  (* bind <name> <value> <env> returns a new environment containing a binding
      of <name> to <value> in addition to all the bindings of <env>. *)

  val bindAll : string list -> 'a list -> 'a env -> 'a env
  (* bind <names> <values> <env> returns a new environment containing bindings
      of <names> to <values> in addition to all the bindings of <env>. *)

  val make : string list -> 'a list -> 'a env
  (* make <names> <values> <env> returns a new environment containing bindings
      of <names> to <values>. *)

  val lookup : string -> 'a env -> 'a option
  (* lookup <name> <env> returns Some <value> if <name> is bound to <value>
      in <env>; otherwise it returns None. *)

  val merge : 'a env -> 'a env -> 'a env
  (* merge <env1> <env2> returns a new environment that contains all the bindings
      of <env1> and <env2>. If a name is bound in both environments, the
      binding in <env1> has precedence. *)

  val remove : string -> 'a env -> 'a env
  (* remove <name> <env> returns a new environment that contains all the bindings
      of <env> except for any binding for <name>. *)

  val removeAll : string list -> 'a env -> 'a env
  (* remove <names> <env> returns a new environment that contains all the bindings
      of <env> except for any binding for names in <names>. *)

end

Figure 8: An environment signature.
```

The subst function works by performing a near-copy of the given BINDEX abstract syntax tree. There are two places where it does not perform an exact copy:
1. For a variable reference, if the variable name appears in the environment `env`, the variable reference is replaced by the expression bound to the variable name in `env`. Otherwise, the variable reference is copied.

2. For a `bind` expression, the bound variable of the `bind` is always \( \alpha \)-renamed to a fresh variable before substitution is performed on its body. In many cases this does more renaming than is strictly necessary, but it is a simple way to avoid all variable capture problems and makes it unnecessary to check if the `bind`-bound name is bound in `env`. It is possible to write a cleverer version of `subst` that (1) renames variables only when absolutely necessary and (2) performs the renaming and substitution on the `bind` body in a single tree walk rather than two separate tree walks. This is left as an exercise for the reader.

Fig. 9 also defines two auxiliary substitution functions:

```ml
val subst1 : exp -> var -> exp -> exp
  subst1 exp var exp' returns a copy of the expression exp' in which all free occurrences of var have been replaced by exp.

val substAll : exp list -> var list -> exp -> exp
  Assume that exps and vars are equal-length lists of expressions and strings, respectively. Then substAll exps vars exp returns a copy of the expression exp in which all free occurrences of names in vars have been replaced by the corresponding expression in exps.
```

For example:

```
# let testSubst1 e v e' = 
    print_string(expToString(subst1 (stringToExp e) v (stringToExp e')));
val testSubst1 : string -> Bindex.var -> string -> unit = <fun>

# testSubst1 "(+ b c)" "a" "(bind a (* a a) (- a 3))";
(bind a.17 (* (+ b c) (+ b c)) (- a.17 3))- : unit = ()
# testSubst1 "(+ b c)" "a" "(+ (bind b (+ 1 a) (* a b))
  (bind c (* 2 a) (+ a c)))";
(+ (bind b.19 (+ 1 (+ b c)) (* (+ b c) b.19))
  (bind c.18 (* 2 (+ b c)) (+ (+ b c) c.18))
)- : unit = ()

# let testSubstAll es vs e' = 
    print_string(expToString(substAll (List.map stringToExp es) vs (stringToExp e')));
val testSubstAll : string list -> string list -> string -> unit = <fun>

# testSubstAll ["(+ b c)"; "(* a b)"] ["a";"b"]
  "(+ (bind a (/ a b) (- a b))
  (bind b (/ b a) (- b a)))";
(+ (bind a.21 (/ (+ b c) (* a b)) (- a.21 (* a b)))
  (bind b.20 (/ (* a b) (+ b c)) (- b.20 (+ b c)))
)- : unit = ()
```

Note that in Fig. 9, the `rename1` and `renameAll` functions introduced in Sec. 3.3 are easily defined in terms of `subst1` and `substAll`. The fact that these are defined in terms of the general `subst` function and that `rename1` is used in the definition of `subst` may seem unsettling at first. But note that the call to `rename1` in `subst` is called on a strictly smaller subexpression than the expression argument to `subst`. Since `subst` of the whole expression is being defined in terms of the value of `subst` on smaller subexpression, the recursion is well-defined.
(* val subst : exp -> exp Env.env -> exp *)
let rec subst exp env =
  match exp with
  Lit i -> exp
  | Var v -> (match Env.lookup v env with
      Some e -> e
    | None -> exp)
  | BinApp(rator,rand1,rand2) ->
      BinApp(rator, subst rand11 env, subst rand2 env)
  | Bind(name,defn,body) ->
      (* Take the simple approach of renaming every name.
          With more work, we could avoid renaming
          unless absolutely necessary. *)
      let name’ = StringUtils.fresh name in
      Bind(name’, subst defn env,
         subst (rename1 name name’ body) env)
      (* note: could be cleverer and do a single
         substitution/renaming *)

(* val subst1 : exp -> var -> exp -> exp *)
(* subst1 <exp> <var> <exp’> substitutes <exp> for <var> in <exp’> *)
and subst1 newexp name exp = subst exp (Env.make [name] [newexp])

(* val substAll : exp list -> var list -> exp -> exp *)
(* subst <exps> <vars> <exp> substitutes <exps> for <vars> in <exp> *)
and substAll newexps names exp = subst exp (Env.make names newexps)

(* val rename1 : var -> var -> exp -> exp *)
(* rename <oldName> <newName> <exp> renames <oldName> to <newName> in <exp> *)
and rename1 oldname newname exp = subst1 (Var newname) oldname exp

(* val renameAll : string list -> var list -> exp -> exp *)
(* rename <oldNames> <newNames> <exp> renames <oldNames> to <newNames> in <exp> *)
and renameAll oldnames newnames exp =
    substAll (List.map (fun s -> Var s) newnames) oldnames exp

Figure 9: Substitution and renaming functions for BINDEX.
4 A Substitution Model Interpreter for BINDEX

With a substitution function in hand, we can implement a substitution model interpreter for BINDEX (Fig. 10). This is similar to the INTEX interpreter except:

- Rather than passing the argument list as an argument to eval, the argument integers are substituted for the formal parameters using substAll. Note that the integers must be converted to expressions (via the Lit constructor) before the substitutions can take place.

- The eval function does not need any argument other than the expression being evaluated, so it has type exp -> int.

- A bind node is evaluated by (1) first evaluating the definition expression to an integer and then (2) evaluating the result of substituting this integer for the bind-bound variable name in the body of the bind expression.

- How is a variable reference node evaluated? In the BINDEX substitution model, all variable references should be replaced by integer literals before they are encountered by eval. If a variable reference is encountered by eval, it must be the case that it is a free variable in the program — i.e., it is an unbound variable error.

5 An Environment Model Interpreter for BINDEX

An alternative interpretation strategy for BINDEX expressions is the environment model. In this strategy, substitutions of integers for variables are not performed eagerly but are delayed by remembering them in environments. The eval function is modified to accept a second argument that is an environment of all delayed substitutions, and the substitutions are only performed when a variable reference is reached.

An interpreter based on this strategy is presented in Fig. 11. Like the argument list in the INTEX interpreter, an initial environment associating the formal parameter names and the actual integer arguments is passed down the abstract syntax tree of the program body as part of evaluation. Unlike the INTEX argument list, however, the BINDEX environment is modified as it “flows” down the tree at every bind node, which adds a new binding to the environment used for the body of the bind.
module BindexSubstInterp = struct
  open Bindex
  open List

  exception EvalError of string

  (* val run : Bindex.pgm -> int list -> int *)
  let rec run (Pgm(fmls,body)) ints =
    let flen = length fmls
    and ilen = length ints
    in
    if flen = ilen then
      eval (substAll (map (fun i -> Lit i) ints) fmls body)
    else
      raise (EvalError ("Program expected " ^ (string_of_int flen) ^ " arguments but got " ^ (string_of_int ilen)))

  (* val eval : Bindex.exp -> int *)
  and eval exp =
    match exp with
      Lit i -> i
    | Var name ->
      raise (EvalError("Unbound variable: " ^ name))
    | BinApp(rator,rand1,rand2) ->
      binApply rator (eval rand1) (eval rand2)
    | Bind(name,defn,body) ->
      eval (subst1 (Lit (eval defn)) name body)

  (* val binApply : Bindex.binop -> int -> int -> int *)
  (* The definition of this function is unchanged from Intex, and is not shown here. *)

  (* A function for running a program expressed as a string *)
  let runString pgmString args =
    run (sexpToPgm (Sexp.stringToSexp pgmString)) args

  (* A function for running a program contained in a file *)
  let runFile pgmFile args =
    run (sexpToPgm (Sexp.stringToSexp (File.fileToString pgmFile))) args
end

Figure 10: A substitution model BINDEX interpreter.
module BindexEnvInterp = struct

open Bindex
open List

exception EvalError of string

(* val run : Bindex.pgm -> int list -> int *)
let rec run (Pgm(fmls,body)) ints =
  let flen = length fmls
  and ilen = length ints
  in
  if flen = ilen then
    eval body (Env.make fmls ints)
  else
    raise (EvalError ("Program expected " ^ (string_of_int flen) ^ " arguments but got " ^ (string_of_int ilen)))

(* val eval : Bindex.exp -> int Env.env -> int *)
and eval exp env =
  match exp with
  | Lit i -> i
  | Var v ->
    (match Env.lookup name env with
      Some(i) -> i
    | None -> raise (EvalError("Unbound variable: " ^ v)))
  | Bind(name,defn,body) ->
    eval body (Env.bind name (eval defn env) env)
  | BinApp(rator,rand1,rand2) ->
    binApply rator (eval rand1 env) (eval rand2 env)

(* The definitions of the following functions are the same as in BindexSubstInterp and are not shown here. *)
(* val binApply : Bindex.binop -> int -> int -> int *)
(* val runString : string -> int list -> int *)
(* val runFile : string -> int list -> int *)
end

Figure 11: A environment model BINDEX interpreter.